

Excel Funktionen

Grundlagen Wirtschaftsinformatik

WS2020

Wolfgang Murth, Martin Nigitsch

BBAF-VZ/BB (WS2020)

Inhaltsverzeichnis

1	Datumsfunktionen	5
1.1	Allgemeines	5
1.2	Übersicht	8
1.3	Zusammenfassung	10
2	Uhrzeitfunktionen	10
2.1	Allgemeines	10
2.2	Übersicht	10
2.3	Zusammenfassung	11
3	WENN Funktion	12
4	UND Funktion	13
5	ODER Funktion	13
6	Matrixfunktionen	14
6.1	VERGLEICH	14
6.2	INDEX (Matrixversion)	15
6.3	INDEX (Bezugsversion)	16
6.4	Beispiel zu INDEX und VERGLEICH	18
6.5	SVERWEIS	20
6.6	WVERWEIS	21
7	Daten filtern	22
7.1	Autofilter	23
7.2	TEILERGEBNIS	24
7.3	Spezialfilter	26
7.4	Spezialfilter Kriterienbereich	27
7.5	Spezialfilter Vergleichsoperationen	28
7.6	Spezialfilter aktivieren	28
7.7	Spezialfilter mit berechneten Feldern	30
8	Datenbankfunktionen	31
9	Erweiterte WENN Funktionen	32
9.1	SUMMEWENN Funktion	32
9.2	SUMMEWENNS Funktion	33
9.3	ZÄHLENWENN Funktion	34
9.4	ZÄHLENWENNS Funktion	34

Abbildungsverzeichnis

1.1	Datumsdarstellung in Excel	5
1.2	Rechnen mit Datumsangaben	5
1.3	Formatfehler der Datumsberechnung	6
1.4	Tagesdifferenz als Zahl formatiert	6
1.5	Altersdifferenzen, Ausgangslage	6
1.6	Altersdifferenz auf der Zeitachse	7
1.7	Berechnete Altersdifferenzen	7
1.8	Differenzalter auf der Zeitachse	7
1.9	Tatsächliche Altersdifferenz	8
3.1	WENN Funktion	12
3.2	Verschachtelte WENN Funktion	12
4.1	WENN und UND in einer Formel	13
5.1	ODER in einer Formel	13
6.1	VERGLEICH in einem Beispiel	15
6.2	VERGLEICH mit fehlerhaftem Wert	15
6.3	MATRIX Funktion	16
6.4	MATRIX Funktion mit VERGLEICH	16
6.5	INDEX Funktion mit mehreren Bereichen	17
6.6	Verkaufsliste des Obsthändlers	18
6.7	Schnittpunkt mit dem gesuchten Wert	18
6.8	Melonen	19
6.9	März	19
6.10	SVERWEIS zur Rabattfindung	21
6.11	WVERWEIS zur Rabattfindung	22
7.1	Autofilter aus dem Menü auswählen	23
7.2	Eingeschaltetes Autofilter	23
7.3	Inhaltsabhängige Filtermöglichkeiten	24
7.4	TEILERGEBNIS und ausgeblendete Zeilen	25
7.5	Gehaltssumme der Abteilung BH mit TEILERGEBNIS	25
7.6	Spezialfilter mit Datenbereich und Kriterienbereich	26
7.7	Spezialfilter Kriterienbereich	27
7.8	Spezialfilter Menü	29
7.9	Spezialfilter aktivieren	29
7.10	Spezialfilter Fehlermeldung	30
7.11	Spezialfilter mit berechnetem Feld	31
8.1	Datenbankfunktionen	32
9.1	SUMMEWENN Funktion	33
9.2	SUMMEWENNS Funktion	33

9.3	ZÄHLENWENN Funktion	34
9.4	ZAEHLENWENNS Funktion	35

Tabellenverzeichnis

7.1	Funktionsnummern der Aggregatfunktionen	25
7.2	Numerische und Datumsvergleichsoperatoren	28
7.3	Alphanumerische Vergleichsoperatoren	28

1 Datumsfunktionen

1.1 Allgemeines

Für Excel beginnt die Zeitrechnung am 0.1.1900. Das klingt seltsam, hat aber seinen Ursprung in einer längst vergessenen Software, nämlich dem Tabellenkalkulationsprogramm Lotus 123. Dieses Programm hatte einen Bug, denn es sah das Jahr 1900 als Schaltjahr an, was aber nicht stimmt. Um mit dem damaligen Platzhirsch kompatibel zu sein, baute Microsoft absichtlich diesen Fehler ein. Deswegen beginnt die Microsoftsche Zeitrechnung mit dem 0.1. und nicht mit dem 1.1., damit ab dem 1. März 1900 der Kalender wieder stimmt.

Für Excel ist jedes Datum eine Zahl, beginnend mit dem 0.1.1900 mit der Zahl 0. Der 1.1.1900 ist dann 1 und so weiter. Der 1.1.1950 ergibt demnach 18264 und der 11.11.2009 den Wert 40128. Sobald Sie nun eine Zelle als Datum formatieren, weiß Excel, dass es diese Zahl in ein Datumsformat wandeln soll.




Abbildung 1.1: Datumsdarstellung in Excel

Der Vorteil dieser Darstellung ist, dass man mit Daten einfach rechnen kann.




Abbildung 1.2: Rechnen mit Datumsangaben

Der Haken ist nur der, dass Excel in seiner grenzenlosen Intelligenz das Format der Quellzelle übernimmt, welche ein Datum ist. Dadurch erhalten sie nicht die Differenz in Tagen, sondern ein seltsames Datum.


		Differenz in Tagen
01.01.1950	01.11.2009	31.10.1959

Abbildung 1.3: Formatfehler der Datumsberechnung

Sie müssen nun die Zelle auf das Format Standard oder Zahl ändern um das richtige Ergebnis anzuzeigen.

		Differenz in Tagen
01.01.1950	01.11.2009	21854

Abbildung 1.4: Tagesdifferenz als Zahl formatiert



	A	B	C
1	Geburtstag	Aktuelles Datum	Alter
2	01.07.1991	11.11.2009	
3	11.11.1991	11.11.2009	
4	05.12.1991	11.11.2009	
5			

Abbildung 1.5: Altersdifferenzen, Ausgangslage

Berechnen wir nun als Beispiel das Alter von drei Freunden, welche alle im selben Jahr geboren wurden und zwar am 1.7.1991, am 11.11.1991 und am 5.12.1991. Der Zeitpunkt der Berechnung ist am Geburtstag des zweiten, also am 11.11.1991. Wir wissen, dass Excel ein Datum intern als Zahl behandelt. Dadurch kann man ein Datum vom anderen abziehen.

Erstellen wir nun die Berechnungsformel Schritt für Schritt für den ältesten der drei Freunde, welcher am 1.7.1991 geboren wurde.




Abbildung 1.6: Altersdifferenz auf der Zeitachse

Aus der Abbildung 1.6 kann man erkennen, dass zwischen dem 11.11.2009 und dem 1.7.1991 genau 40128-33420 Tage liegen, was 6708 Tagen entspricht. Tragen wir nun einmal diese einfache Formel ein und sehen uns das Ergebnis an.

C2 fx =B2-A2				
	A	B	C	D
1	Geburtstag	Aktuelles Datum	Alter	Alter (Datum)
2	01.07.1991	11.11.2009	6708	13.05.1918
3	11.11.1991	11.11.2009	6575	31.12.1917
4	05.12.1991	11.11.2009	6551	07.12.1917

Abbildung 1.7: Berechnete Altersdifferenzen

In Zelle C2 steht die Formel =B2-A2. In der Spalte C9 ist das Alter in Tagen als Zahl formatiert zu sehen. In der Spalte D dagegen ist der Inhalt der Spalte C als Datum formatiert zu sehen. Es ist erkennbar, dass das Jahr 1917, beziehungsweise 1918, etwas mit dem Alter zu tun haben muss. In der unteren Abbildung ist die Erklärung dafür. Nehmen Sie die Differenz der beiden Zeitpunkte und betrachten Sie diese vom Anbeginn der Microsoftschen Zeitrechnung aus.




Abbildung 1.8: Differenzalter auf der Zeitachse

Um das eigentliche Alter in Jahre zu erhalten sind jetzt noch 2 Schritte notwendig

- Mittels der Funktion JAHR() aus dem Differenzdatum nur die Jahreszahl extrahieren.
Mit =JAHR(B2-A2) erhalten Sie 1917, beziehungsweise 1918

- Den Beginn der Microsoftschen Zeitrechnung, also 1900, abziehen Mit =JAHR(B2-A2)-1900 erhalten Sie 17, beziehungsweise 18

Damit scheint das Ergebnis erreicht zu sein. Leider nur scheinbar, denn am Tag des Geburtstages wird ein falsches Alter angezeigt. Um das zu korrigieren, addiert man einfach einen Tag zur Differenz zwischen beiden Daten. Das Ergebnis sieht dann wie folgt aus.

	A	B	C	D
1	Geburtstag	Aktuelles Datum	Alter	
2	01.07.1991	11.11.2009	18	
3	11.11.1991	11.11.2009	18	
4	05.12.1991	11.11.2009	17	

Abbildung 1.9: Tatsächliche Altersdifferenz

1.2 Übersicht

HEUTE()	Liefert das aktuelle Datum Z.B. HEUTE() ⇒ 01.09.2015
JETZT()	Liefert das aktuelle Datum samt Uhrzeit Z.B. JETZT ⇒ 21.02.2014 12:00:23
TAG(<Datum>)	Liefert den Tag eines Datums als Ganzzahl Z.B. TAG("21.02.2014") ⇒ 21
MONAT(<Datum>)	Liefert das Monat eines Datums als Ganzzahl Z.B. MONAT("21.02.2014") ⇒ 2
JAHR(<Datum>)	Liefert das Jahr eines Datums als Ganzzahl Z.B. JAHR("21.02.2014") ⇒ 2014
DATUM(<Jahr>; <Monat>; <Tag>)	Erstellt aus 3 Ganzzahlen ein Datum Z.B. Datum(21; 2; 2014) ⇒ 41691 Diese Funktion ist eine der wichtigsten Datumsfunktionen, weil Sie eine beliebige Addition oder Subtraktion von Jahren, Monaten und Tagen ermöglicht und daraus ein neues Datum erstellen kann.
MONATSENDE(<Datum>, <Monat>)	Liefert das Monatsende als Datum (Zahl) zurück, das eine bestimmte Anzahl von Monaten vor bzw. nach einem Datum liegt. Z.B. ("21.02.2014"; 3) ⇒ 31.05.2014

WOCHENTAG(<i>⟨Datum⟩</i> ; <i>⟨Typ⟩</i>)	Liefert den Wochentag eines Datums als Ganzzahl zurück, wobei die Typangabe (Default = 1) entscheidet, welcher Wochentag mit welcher Zahl beginnt. (bei Default gilt 1=So, 2=Mo, 3=Di,...) Z.B. (z.B. ("21.02.2014"; 1) ⇒ 6
TEXT(<i>⟨Datum⟩</i> ; " <i>⟨Format⟩</i> ")	Gibt ein Datum entsprechend dem ausgewählten Format an ¹ . Z.B. TEXT("21.02.2014"; "TTT") ⇒ Fr
DATEDIF(<i>⟨Startdatum⟩</i> ; <i>⟨Enddatum⟩</i> ; <i>⟨Einheit⟩</i>)	Liefert die Differenz als Ganzzahl zwischen zwei Datums- werten. Als Einheit kann man "Y", "M" oder "D" angeben. Es ist zu beachten, dass das Startdatum immer vor dem End- datum liegen muss. Z.B. DATEDIF(HEUTE(); "24.12.2014"; "D") ⇒ 306
ARBEITSTAG(<i>⟨Startdatum⟩</i> ; <i>⟨Tage⟩</i> ; <i>⟨Freitage⟩</i>)	Liefert das Datum des nächsten (oder zurückliegenden) Arbeitstages durch Addition oder Subtraktion von Tagen zu einem Startdatum unter Berücksichtigung eventueller Freitage (Matrix).
NETTOARBEITSTAGE(<i>⟨Beginndatum⟩</i> ; <i>⟨Enddatum⟩</i> ; <i>⟨Freitage⟩</i>)	Liefert die Anzahl der Arbeitstage zwischen einem An- fangsdatum und einem Enddatum unter Berücksichtigung eventueller Freitage (Matrix).
EDATUM(<i>⟨Ausgangsdatum⟩</i> ; <i>⟨Monate⟩</i>)	Liefert das Datum (Zahl) zurück, das eine bestimmte An- zahl von Monaten vor bzw. nach einem Ausgangsdatum liegt. Z.B. EDATUM("29.02.2016"; -12) ⇒ 28.02.2015
KALENDERWOCHE(<i>⟨Datum⟩</i> ; <i>⟨Typ⟩</i>)	Liefert die Kalenderwoche eines Datums als Ganzzahl zu- rück. Typ 1: Woche mit dem 1.1. ist die 1. Kalenderwoche (ame- rikanisches System) Typ 21: Woche, die den 1. Donnerstag im Jahr hat, ist die Kalenderwoche 1 oder die den 4.1. beinhaltet (entspricht ISO EU-Standard).

¹<https://support.office.com/de-at/article/TEXT-Funktion-20d5ac4d-7b94-49fd-bb38-93d29371225c?ui=de-DE&rs=de-AT&ad=AT>

1.3 Zusammenfassung

- In Excel werden Datumswerte als Ganzzahl interpretiert. Das Standard-Datumsystem basiert darauf, dass der 1.1.1900 der Zahl 1 entspricht. Das letzte interpretierbare Datum ist der 31.12.9999, was der Zahl 2.958.465 entspricht.
- Auf Grund des Jahrtausendwechsels sind Jahreszahlen im vierstelligen Format anzugeben. Die zweistelligen Jahreszahlen 00-29 werden als 2000 bis 2029 interpretiert, 30 bis 99 hingegen als 1930 bis 1999.
- Da In Excel Datums- und Uhrzeitangaben als Zahlen repräsentiert werden, kann man daher auch mittels mathematischer Operatoren, wie Addition und Subtraktion, mit Datumswerten rechnen.

2 Uhrzeitfunktionen

2.1 Allgemeines

Für die Uhrzeit, oder generell Zeitdarstellung, wird einfach der Nachkommateil des Datums herangezogen. 0,5 entspricht hier genau 12 Uhr, 0,75 wäre 18 Uhr. Fügt man nun Datum und Uhrzeit zusammen, so erhält man mit 40128,5 den 11.11.2009 und 12 Uhr Mittags.

Eine Funktion, welche für Zeitberechnungen immer wieder verwendet wird, ist `Zeit`, mit der man getrennte Stunden, Minuten und Sekunden zu einem Zeitwert zusammenführt.

2.2 Übersicht

<code>STUNDE(<Uhrzeit>)</code>	Liefert die Stunde einer Uhrzeit als Ganzzahl Z.B. <code>STUNDE("12:34:23")</code> \Rightarrow 12
<code>MINUTE(<Uhrzeit>)</code>	Liefert die Minute einer Uhrzeit als Ganzzahl Z.B. <code>MINUTE("12:34:23")</code> \Rightarrow 34
<code>SEKUNDE(<Uhrzeit>)</code>	Liefert die Sekunde einer Uhrzeit als Ganzzahl Z.B. <code>SEKUNDE("12:34:23")</code> \Rightarrow 23
<code>ZEIT(<Stunde>; <Minute>; <Sekunde>)</code>	Erstellt aus drei Ganzzahlen eine Uhrzeit Z.B. <code>ZEIT(12; 34; 23)</code> \Rightarrow 12:34:23 \Rightarrow 0,523877314814815



Die ZEIT() Funktion eignet sich wie die DATUM()Funktion, um aus Addition oder Subtraktion von Stunden, Minuten und Sekunden eine neue Uhrzeit zu bekommen. Es ist allerdings zu beachten, dass bei einem Überschreiten der 24 Stunden die Zeitfunktion allfällige Tage abschneidet. So ergibt ZEIT(27; 0; 0) nicht den Wert 1,125 sondern 0,125, was der Uhrzeit von 3 Uhr entspricht.

Wenn als Beispiel zu dem Zeitpunkt 15.03.2014 11:00:00 eine Zeit von 30 Stunden, 20 Minuten und 10 Sekunden addiert werden soll bringt die Formel

$$= "15.3.2014 11:00:00" + ZEIT(30;20;10)$$

das nicht richtige Ergebnis: 15.3.2014 17:20:10!



Um dieses Problem zu lösen, ist es ratsam, prinzipiell für Zeitoperationen Stunden, Minuten und Sekunden in Bruchteile eines Tages umzurechnen. Also lautet die Formel für die korrekte Berechnung

$$= "15.03.2014 11:00:00" + 30/24 + 20/24/60 + 10/24/60/60$$

welche das korrekte Ergebnis 16.03.2014 17:20:10 liefert.

2.3 Zusammenfassung

- Uhrzeiten werden als Bruchteil von 1 angesehen. So entspricht die Zahl 0,5 der Uhrzeit 12:00 oder 0,375 entspricht 09:00. Der Termin 05.03.2014 11:15 entspricht daher dem Zahlenwert 41703,46875.
- Da In Excel Datums- und Uhrzeitangaben als Zahlen repräsentiert werden, kann man daher auch mittels mathematischer Operatoren, wie Addition und Subtraktion, mit Datumswerten rechnen.

3 WENN Funktion

WENN(⟨Bedingung⟩; [⟨Dann_Wert⟩]; [⟨Ansonsten_Wert⟩])

B2		fx		=WENN(A2>1000;"Große Zahl";"Kleine Zahl")			
	A	B	C	D	E	F	G
1							
2		10	Kleine Zahl				
3		10000	Große Zahl				
4							

Abbildung 3.1: WENN Funktion

Die WENN Funktion liefert in einfachster Form einen Wert zurück. Wenn die ⟨Bedingung⟩ als WAHR ausgewertet wird, liefert sie den ⟨Dann_Wert⟩, wird sie als FALSCH ausgewertet liefert sie den ⟨Ansonsten_Wert⟩.

Jede WENN Funktion, auch eine verschachtelte, wird automatisch beendet, wenn ein Zweig durchlaufen wird. WENN Funktionen können bis zu 64-fach ineinander verschachtelt werden. Achten Sie bei verschachtelten WENN Funktionen, dass Ihre numerischen Abgrenzungen in den Bedingungen einer logischen Abfolge gleichen.

Im untenstehenden Beispiel ist eine verschachtelte WENN Funktion dargestellt. Ist das Alter mindestens 65 Jahre, dann wird „ALT“ angezeigt, ist es das nicht, wird danach geprüft, ob das Alter zumindest 35 Jahre ist. Ist das zutreffend, wird „MITTEL“ angezeigt. Ist die Person unter 35 Jahren, dann wird „JUNG“ ausgegeben.

C3		fx		=WENN(B3=65;"ALT";WENN(B3>=35;"MITTEL";"JUNG"))					
	A	B	C	D	E	F	G	H	I
1									
2	Person	Alter	Einstufung						
3	Karl	67	MITTEL						
4	Eva	35	MITTEL						
5	Maria	27	JUNG						
6	Werner	17	JUNG						
7									

Abbildung 3.2: Verschachtelte WENN Funktion



Grundsätzlich gilt die Merkregel:

Numerische Grenzen von großem nach kleinem Wert immer mit > oder >=

Numerische Grenzen von kleinem nach großem Wert immer mit < oder <=

4 UND Funktion

UND($\langle \text{Bedingung}_1 \rangle$; $\langle \text{Bedingung}_2 \rangle$; . . . ; $\langle \text{Bedingung}_{255} \rangle$)

Die UND Funktion gibt WAHR zurück, wenn alle Bedingungen WAHR sind.

Beispiele =UND(1>2 ; 2<3) \Rightarrow FALSCH
 =UND(1>0 ; 8/4=2) \Rightarrow WAHR

Die UND Funktion kommt sehr häufig in Formeln mit einer WENN Bedingung vor.

E3 fx =WENN(UND(C3>0;D3>0);"gemischt";WENN(C3>0;"nur Mädchen";"nur Buben"))													
	A	B	C	D	E	F	G	H	I	J	K	L	M
1													
2		Klasse	Mädchen	Buben	Art								
3		3A	30	0	nur Mädchen								
4		7B	15	9	gemischt								
5		5C	18	10	gemischt								
6		1D	0	32	nur Buben								
7													

Abbildung 4.1: WENN und UND in einer Formel

5 ODER Funktion

ODER($\langle \text{Bedingung}_1 \rangle$; $\langle \text{Bedingung}_2 \rangle$; . . . ; $\langle \text{Bedingung}_{255} \rangle$)

Die ODER Funktion gibt WAHR zurück wenn zumindest eine der Bedingungen WAHR ergibt.

Beispiele =ODER(1<0 ; 8/4=2) \Rightarrow WAHR
 =ODER(1=2 ; 4/0,5=2) \Rightarrow FALSCH

B3 fx =ODER(1<0;8/4=2)					
	A	B	C	D	E
1					
2					
3		WAHR			=ODER(1<0;8/4=2)
4		FALSCH			=ODER(1=2;4/0,5=2)
5					

Abbildung 5.1: ODER in einer Formel

6 Matrixfunktionen

Matrixfunktionen bieten uns die Möglichkeit, nach bestimmten Kriterien Werte aus Listen oder Bereichen zu entnehmen, ihre Positionen zu bestimmen oder bieten uns Hilfestellungen dafür an.

6.1 VERGLEICH

VERGLEICH(*Suchkriterium*; *Suchmatrix*; [*Vergleichstyp*])

Die VERGLEICH Funktion wird dann verwendet, wenn man nicht den Wert aus einer Spalte oder Zeile haben will, sondern die Position bezogen auf die Suchmatrix.

- Suchkriterium* Wert nach dem in der *Suchmatrix* gesucht werden soll
- Suchmatrix* ist entweder ein Zeilenbereich, z.B. "A2:H2", oder ein Spaltenbereich, z.B. "A2:A6", in dem gesucht werden soll.
- Vergleichstyp*
- | | |
|-------------|--|
| 1 (default) | Sucht nach dem größtem Wert, der kleiner oder gleich dem Suchkriterium ist. Die Suchmatrix muss daher zwingend <i>aufsteigend</i> sortiert sein. |
| 0 | Sucht exakt das Suchkriterium in der Suchmatrix. Das Suchkriterium muss in der Suchmatrix vorkommen, die Suchmatrix muss aber nicht sortiert sein. |
| -1 | Sucht nach dem kleinsten Wert, der größer oder gleich dem Suchkriterium ist. Die Suchmatrix muss daher zwingend <i>absteigend</i> sortiert sein. |

Im folgenden Beispiel ist das Suchkriterium der Wert 245. VERGLEICH sucht nun im angegebenen Suchbereich jene Zeile, in der der gesuchte Wert größer gleich dem Wert in der Zeile, aber kleiner als der Wert in der nächsten Zeile ist. In diesem Falle, ist es hier die Zeile mit dem Wert 200. Der ist ja kleiner als unser gesuchter Wert. Und der Wert in der nächsten Zeile ist schon größer als unser gesuchter Wert. Der gesuchte Wert ist also in der zweiten Zeile des Suchbereiches und daher liefert die Formel den Wert 2.

E5			fx =VERGLEICH(E2;\$B\$2:\$B\$7;1)						
	A	B	C	D	E	F	G	H	I
1									
2	Suchmatrix (B2:B7)	100		Suchkriterium:	245				
3		200							
4		300							
5		400		Ergebnis:	2				=VERGLEICH(E2;\$B\$2:\$B\$7;1)
6		500							
7		600							

Abbildung 6.1: VERGLEICH in einem Beispiel



Wenn der Suchwert kleiner als der kleinste Wert in der Suchmatrix ist, dann wird ein Fehler #NV retourniert. Fehler wie #NV, #Wert!, #Bezug!, #Div/0!, #Num!, #Name! oder #Null! können mit der Funktion WENNFEHLER(⟨Wert⟩; ⟨Wert falls Fehler⟩) abgefangen werden.

E5		fx =WENNFEHLER(VERGLEICH(E2;\$B\$2:\$B\$7;1);"Wert < 100")									
	A	B	C	D	E	F	G	H	I	J	K
1											
2	Suchmatrix (B2:B7)	100		Suchkriterium:	50						
3		200									
4		300									
5		400		Ergebnis:	Wert < 100						
6		500									
7		600									

Abbildung 6.2: VERGLEICH mit fehlerhaftem Wert

6.2 INDEX (Matrixversion)

INDEX(⟨Matrix⟩; ⟨Zeile⟩; ⟨Spalte⟩)

Die INDEX Funktion liefert aus einer Matrix genau jenen Wert, der im angegebenen Schnittpunkt einer Zeile und einer Spalte steht. Besteht die Matrix aus nur einem Zeilenbereich oder einem Spaltenbereich kann das Argument der Zeile bzw. Spalte weggelassen werden. Im einfachsten Anwendungsfall, der sehr selten vorkommt, weiß man, in welcher Zeile der Matrix der gewünschte Spaltenwert zu suchen ist. In der Abbildung 6.3 wird der Preis der Zitrone gewünscht, welche in der dritten Zeile der Matrix steht.

B8		fx		=INDEX(A2:C6;3;2)	
	A	B	C	D	E
1	Obst	Preis	Anzahl		
2	Apfel	0,69	40		
3	Banane	0,34	38		
4	Zitrone	0,55	15		
5	Orange	0,25	25		
6	Birne	0,59	40		
7					
8	Zitronenpreis	0,55			=INDEX(A2:C6;3;2)
9					

Abbildung 6.3: MATRIX Funktion

Normalerweise kennt man entweder die Spalte nicht, die Zeile nicht oder auch beides gemeinsam nicht. Wenn das obige Beispiel derart erweitert wird, dass auch die Obstsorte wählbar ist, dann würde die Formel folgendermaßen aussehen.


	B9			<i>f_x</i>	=INDEX(A2:C6;VERGLEICH(B8;\$A\$2:\$A\$6;0);2)		
	A	B	C	D	E	F	G
1	Obst	Preis	Anzahl				
2	Apfel	0,69	40				
3	Banane	0,34	38				
4	Zitrone	0,55	15				
5	Orange	0,25	25				
6	Birne	0,59	40				
7							
8	Artikel	Zitrone					
9	Gesamtpreis	0,55		=INDEX(A2:C6;VERGLEICH(B8;\$A\$2:\$A\$6;0);2)			
10							

Abbildung 6.4: MATRIX Funktion mit VERGLEICH



Es ist wichtig, dass man eventuelle Spalten- oder Zeilenbeschriftungen *nicht* zum Matrixbereich hinzuzählt. Der Matrixbereich ist nur der Wertebereich.

6.3 INDEX (Bezugsversion)

INDEX(<Matrix>; <Zeile>; <Spalte>; [*<Bereich>*])

Liefert den Bezug jener Zelle, auf die sich Zeile und Spalte in dem angegebenen Bezug (=Zellbereich) beziehen. Für den Parameter <Bezug> können auch mehrere, voneinander unabhängige Zellbereiche durch () eingeschlossen, angegeben werden.

Bereich Falls mehrere Bezüge angegeben werden, bezieht sich die Bereichsnummer beginnend mit 1, dem Defaultwert, auf den jeweiligen Bezugsbereich.

	A	B	C	D	E	F	G	H	I	J
1			Gütekategorie 1		Gütekategorie 2		Gütekategorie 3			
2	Ware	Code	Preis	Anzahl	Preis	Anzahl	Preis	Anzahl		
3	Apfel	2BD12	0,69	40	0,55	55	0,34	77		
4	Banane	2LI60	0,34	38	0,20	67	0,11	123		
5	Zitrone	1QV11	0,55	15	0,46	40	0,33	213		
6	Orange	2JF87	0,25	25	0,25	10	0,16	45		
7	Birne	2SG42	0,59	40	0,40	78	0,23	56		
8	Mandel	1HR99	2,80	10	2,30	34	1,76	54		
9	Cashewkern	3RA76	3,55	16	3,20	43	2,56	20		
10	Erdnuss	2RA20	1,25	20	1,10	13	0,89	33		
11	Walnuss	3NS73	1,75	12	1,56	54	1,12	100		
12										
13	Gütekategorie	3								
14	Code	2JF87								
15	Ware	Orange	=INDEX(A3:B11;VERGLEICH(B6;B3:B11;0);1)							
16	Preis	0,16	=INDEX((\$C\$3:\$D\$11;\$E\$3:\$F\$11;\$G\$3:\$H\$11);VERGLEICH(B14;\$B\$3:\$B\$11;0);1;\$B\$13)							
17	Lagerstand	45	=INDEX((\$C\$3:\$D\$11;\$E\$3:\$F\$11;\$G\$3:\$H\$11);VERGLEICH(B14;\$B\$3:\$B\$11;0);2;\$B\$13)							
18										

Abbildung 6.5: INDEX Funktion mit mehreren Bereichen

Schauen wir uns die Formel für den Lagerstand in Zelle B17 näher an und teilen die einzelnen Parameter zur näheren Erklärung in ihre Einzelteile auf

=INDEX((\$C\$3:\$D\$11;\$E\$3:\$F\$11;\$G\$3:\$H\$11);VERGLEICH(B14;\$B\$3:\$B\$11;0);2;\$B\$13)

(\$C\$3:\$D\$11;\$E\$3:\$F\$11;\$G\$3:\$H\$11)

Der erste Parameter umfasst die drei verschiedenen Bereiche mit Preis und Lagerstand.

VERGLEICH(B14;\$B\$3:\$B\$11;0)

Danach wird über die VERGLEICH Funktion die Zeile ermittelt, welche den gewünschten Wert enthält. In diesem Fall wird nach dem Code, welcher in B14 angegeben ist, im Bereich \$B\$3:\$B\$11 gesucht. Und das auf genaue Übereinstimmung.

2

Da der Lagerstand gewünscht ist, soll aus der Matrix der Wert der zweiten Spalte zurückgegeben werden.

\$B\$13

Im letzten Parameter wird nun bestimmt, aus welchem der im ersten Parameter angegebenen Bereiche der Wert genommen wird. Im obigen Beispiel bezeichnet die 3 der Güteklasse auch den dritten Bereich



Es ist absolut wichtig, dass mehrere Bereiche immer durch Klammern () umgeben sind, da sonst ein Fehler ausgegeben wird.

6.4 Beispiel zu INDEX und VERGLEICH

Die beiden Excel Funktionen INDEX und VERGLEICH arbeiten sehr oft zusammen und daher werden sie hier gemeinsam anhand eines Beispiels eines Obsthändlers erklärt, der wissen will wieviele Melonen im März verkauft wurden.

Wieviele Melonen wurden im März verkauft?

	Januar	Februar	März
Äpfel	12	25	88
Melonen	6	33	65
Ananas	12	11	43
Birnen	26	33	99

Suche	
Obst	Melonen
Monat	März

Ergebnis mit Einzelschritten		
	Vergleich	Index Ergebnis
Obst	2	65
Monat	3	

Zusammengefasstes Ergebnis	
65	

Abbildung 6.6: Verkaufsliste des Obsthändlers

Sieht man sich die Tabelle im Bereich C5:E8 an, kann man erkennen, dass der gesuchte Wert der Schnittpunkt der Zeile 6, für die Melonen, und der Spalte E, für den Monat März, ist. Auf die Matrix bezogen ist der gesuchte Wert in der zweiten Zeile und der dritten Spalte. Die Spalten- und Reihenbezeichnung darf hier, wie bereits oben erwähnt, nicht miteinbezogen werden.

	Januar	Februar	März
Äpfel	12	25	88
Melonen	6	33	65
Ananas	12	11	43
Birnen	26	33	99

Abbildung 6.7: Schnittpunkt mit dem gesuchten Wert

Jetzt bleibt noch die Frage wie man die richtige Zeile und Spalte findet. Das einfachste ist, im Bereich B5:B8 nach dem Wort Melonen und im Bereich C4:E4 nach März zu suchen. Dafür wird die Excel

Funktion VERGLEICH angewendet, welche in einem Suchbereich nach dem Suchkriterium gesucht. Man erhält die relative Position des gesuchten Elementes im Suchbereich.

=VERGLEICH(<Suchkriterium>; <Suchbereich>; <Vergleichstyp>)

Zuerst wird in C16 die Zeile der Matrix B5:B8 mit dem Text Melonen ermittelt.

=VERGLEICH(C11; B5:B8; 0)

Es wird also der Inhalt von C11 im Bereich B5:E8 gesucht. Der dritte Parameter der Vergleichfunktion bestimmt die Art des Vergleiches. Wenn er 0 ist, dann wird nach einer genauen Übereinstimmung gesucht. Wie zu erwarten war, erhält man den Wert 2, was nichts anderes bedeutet, dass in der zweiten Zeile des Suchbereiches der gesuchte Wert zu finden ist.

Äpfel	12	25
Melonen	6	33
Ananas	12	11
Birnen	26	33

Suche	
Obst	Melonen
Monat	März

Ergebnis mit Einzelsch		
	Vergleich	Inc
Obst	=VERGLEICH(C11;B5:B8;0)	
Monat	3	

Abbildung 6.8: Melonen

Genauso wird nun beim Monat verfahren, dessen Spalte in C17 ermittelt wird. =VERGLEICH(C12; C4:E4; 0) Das Ergebnis ist, dass der Monat März in der dritten Zeile des Bereiches C4:E4 zu finden ist.

	Januar	Februar	März
Äpfel	12	25	88
Melonen	6	33	65
Ananas	12	11	43
Birnen	26	33	99

Suche	
Obst	Melonen
Monat	März

Ergebnis mit Einzelschritten		
	Vergleich	Index Ergebn
Obst	2	65
Monat	=VERGLEICH(C12;C4:E4;0)	

Abbildung 6.9: März

Nachdem nun sowohl die Spalte als auch die Zeile des Datenbereiches bekannt ist, kann man mit der Funktion INDEX den gesuchten Wert aus der Matrix ermitteln.

=INDEX(<Matrix>; <Zeile>; <Spalte>)

Überträgt man nun die ermittelte Zeile und Spalte in diese Formel, so sieht das Ergebnis so aus

=INDEX(C5:E8; C16; C17)

Die Zwischenschritte werden normalerweise nicht gemacht, sondern es wird die komplette Formel in einer Zelle verwendet. Dann sieht das Ergebnis so wie in D21 aus

=INDEX(C5:E8;VERGLEICH(C11;B5:B8;0);VERGLEICH(C12;C4:E4;0))

6.5 SVERWEIS

SVERWEIS(*Suchkriterium*; *Matrix*; *Spaltenindex*; [*Bereichsverweis*])

Diese Funktion ist eine Art Kombination aus VERGLEICH und INDEX in der Matrixversion. Sie sucht ähnlich wie VERGLEICH innerhalb einer Spalte einen Suchwert und gibt aus einer Matrix den Wert im Schnittpunkt der so gefundenen Zeile mit einer eingegebenen Spaltenzahl zurück.

<i>Suchkriterium</i>	Dieser Wert wird in der in der ersten Spalte, also die sich ganz links befindet, der Suchmatrix gesucht. Genauso wie bei der VERGLEICH Funktion gibt es einen #NV Fehler, wenn der Suchwert kleiner als der kleinste Wert in der Suchspalte ist.
<i>Matrix</i>	Ein Zellbezug von mindestens zwei Spalten
<i>Spaltenindex</i>	Eine Ganzzahl, welche jene Spalte der Matrix angibt, deren Wert zurückgegeben werden soll. Wenn eine Spalte < 1 gewählt wird, gibt es den Fehler #Wert!, ist die gewünschte Spalte größer als die Spaltenzahl der Matrix, dann gibt es einen Fehler #Bezug!.
<i>Bereichsverweis</i>	WAHR (default) Sucht nach dem größtem Wert, der kleiner oder gleich dem Suchkriterium ist. Die erste Spalte der Suchmatrix muss aufsteigend sortiert sein.
	FALSCH Sucht eine genaue Übereinstimmung. Hier muss die erste Spalte der Matrix <i>nicht</i> sortiert sein.

Nehmen wir an, dass Kunden auf Grund ihrer Umsätze am Ende des Jahres einen Rabatt bekommt. Es wird also für jeden Kunden einzeln in der Rabatttabelle, welche hier die Matrix ist, nachgesehen, welchen Rabatt der Kunde bekommt.

	C3		f _x	=SVERWEIS(B3;\$G\$3:\$H\$7;2)					
	A	B	C	D	E	F	G	H	
1									
2	Kunde	Umsatz	Rabatt	Betrag			Umsatz	Rabatt	
3	Maier GmbH	13.456,00 €	3%	403,68 €	=SVERWEIS(B3;\$G\$3:\$H\$7;2)		0,00 €	0%	
4	Striezelkönig	45.213,00 €	5%	2.260,65 €			5.000,00 €	2%	
5	Mamas Bester	72.221,00 €	10%	7.222,10 €			10.000,00 €	3%	
6	Tauchsport Foobar	3.567,00 €	0%	0,00 €			20.000,00 €	5%	
7	Reitschule Hofer	7.894,00 €	2%	157,88 €			50.000,00 €	10%	
8	Planenbau Seifried	33.812,00 €	5%	1.690,60 €					
9									
10			Suchkriterium						
11			Matrix						
12									

Abbildung 6.10: SVERWEIS zur Rabattfindung

Bei der Reitschule Hofer beträgt der Jahresumsatz 7.894 Euro. Dieser Wert wird herangezogen, um in der Umsatzspalte der Matrix, dem Bereich F3:G7, die Zeile zu finden die größer gleich dem Wert in der Zeile, aber kleiner als der Wert in der nächsten Zeile ist. 7.894 ist größer als 5.000, aber kleiner als 10.000. Daher ist diese Zeile die gesuchte. Nun wollen wir nur den Wert aus der zweiten Spalte, die ja den Rabattbetrag festlegt.



Man kann sich die Funktion des SVERWEIS ganz einfach durch „Was suche ich wo und die wievielte Spalte will ich als Ergebnis haben.“ merken.



Beim Verwenden von SVERWEIS wird gerne übersehen, dass man den Matrixbereich in der Formel fixieren sollte. Also statt F4:G7 sollte \$F\$4:\$G\$7 verwendet werden. Dadurch wird der Matrixbereich beim Kopieren der Formel in andere Zellen nicht verändert.

6.6 WVERWEIS

Der WVERWEIS funktioniert genauso wie der SVERWEIS, nur dass die Matrix nicht zeilenorientiert, sondern spaltenorientiert aufgebaut ist.

SVERWEIS(<Suchkriterium>; <Matrix>; <Zeilenindex>; [<Bereichsverweis>])

Das Beispiel aus dem Abschnitt mit dem SVERWEIS würde also so aussehen.

	C7		f_x	=WVERWEIS(B7;B\$2:\$F\$3;2)		
	A	B	C	D	E	F
1						
2	Umsatz	0,00 €	5.000,00 €	10.000,00 €	20.000,00 €	50.000,00 €
3	Rabatt	0%	2%	3%	5%	10%
4						
5						
6	Kunde	Umsatz	Rabatt	Betrag		
7	Maier Gmbh	13.456,00 €	3%	403,68 €	=WVERWEIS(B7;B\$2:\$F\$3;2)	
8	Striezelkönig	45.213,00 €	5%	2.260,65 €		
9	Mamas Bester	72.221,00 €	10%	7.222,10 €		
10	Tauchsport Foobar	3.567,00 €	0%	0,00 €		
11	Reitschule Hofer	7.894,00 €	2%	157,88 €		
12	Planenbau Seifried	33.812,00 €	5%	1.690,60 €		
13						
14			Suchkriterium			
15			Matrix			

Abbildung 6.11: WVERWEIS zur Rabattfindung

7 Daten filtern

Aus Tabellen (Listen) können Daten nach bestimmten Kriterien gefiltert werden. Alle Zeilen, welche nicht dem Filterkriterium entsprechen werden dabei ausgeblendet.

Wenden Sie mehrere Filter an, so ist deren Wirkung kumulativ. Das bedeutet, dass der zweite Filter nur noch mit den Daten arbeitet, welche das erste Filter bereitstellt.

Generell ist zu beachten, dass in den Spalten immer nur ein Datentyp vorkommen darf, wie es auch relationale Datenbanken vorschreiben. Also sollen in einer Spalte z.B. immer nur numerische Werte oder Datumswerte vorkommen. Es soll also nicht sein, dass numerische mit nichtnumerischen Werten gemischt werden. Das Autofilter bietet nämlich vom Datentyp abhängige Filter automatisch an. Werden in einer Spalte gemischte Datentypen verwendet, dann wird das Filter für den am häufigsten vorkommenden Datentyp angeboten.

Wird ein Filter auf eine Spalte angewendet, für die bereits ein Filter existiert, wird der alte Filter automatisch gelöscht.

Grundsätzlich gibt es in Excel zwei verschiedene Filterarten

- Autofilter
- Spezialfilter

7.1 Autofilter

Bevor man das Autofilter einschaltet, sollte man eine Zelle im Datenbereich auswählen. Excel kann dann, wenn man die Überschriften anders als die Datenspalten formatiert hat, automatisch den kompletten Datenbereich auswählen. Danach wählt man **Daten** » **Sortieren und Filtern** » **Filter**.




Abbildung 7.1: Autofilter aus dem Menü auswählen

Ist das Autofilter aktiviert, wird neben den Spaltenüberschriften ein kleines Dreieck für die Auswahl der Eingrenzungskriterien angezeigt und das Symbol im Ribbon wird aktiviert angezeigt.




Abbildung 7.2: Eingeschaltetes Autofilter

Autofilter liefern schnell Ergebnisse, wenn es sich um einfache unkomplizierte Bedingungen handelt. Je nach Datentyp in einer Spalte zeigt das Autofilter unterschiedliche Optionen an.




Abbildung 7.3: Inhaltsabhängige Filtermöglichkeiten

7.2 TEILERGEBNIS

Zeilen, welche nicht vom Filter betroffen sind, werden einfach ausgeblendet. Häufig ist man aber nicht an den einzelnen Datensätzen interessiert, sondern an aggregierten, d.h. zusammengefassten, Werten. Dies kann eine Summe, ein Durchschnitt oder Maximum sein. Die von Excel bekannten Funktionen nehmen allerdings keine Rücksicht auf ausgeblendete Zeilen, also sind sie für diese Aufgabenstellung nicht verwendbar. Um aggregierte Werte zu erhalten wird daher die TEILERGEBNIS Funktion verwendet.

TEILERGEBNIS($\langle \text{Funktion} \rangle$; $\langle \text{Bezug} \rangle$; . . . ; $\langle [\text{Bezug}_n] \rangle$)

Die $\langle \text{Funktion} \rangle$ gibt die Art der Aggregation an. Eine Auflistung der Aggregatsfunktionen mit ihren Funktionsnummern zeigt die Tabelle 7.1.

	Funktion	
	berücksichtigt ausgeblendeten Zellen	ignoriert ausgeblendete Zellen
Mittelwert	1	101
Anzahl	2	102
Anzahl2	3	103
Max	4	104
Min	5	105
Produkt	6	106
Stabw	7	107
Stabwn	8	108
Summe	9	109
Varianz	10	110
Varianzen	11	111

Tabelle 7.1: Funktionsnummern der Aggregatfunktionen

In der Abbildung 7.4 kann man sehen, wie sich ausgeblendete Zeilen auswirken. In der Tabelle ist die dritte Zeile ausgeblendet und die Funktion 109 liefert daher den um den Zellwert reduzierten Summenwert gegenüber der Funktion 9. In diesem Beispiel ist der Zellinhalt 3, die Summe also 12. Abbildung 7.5 zeigt weiters die Gehaltssumme der Abteilung BH mit Hilfe des Autofilter und TEILERGEBNIS.

	A	B	C	D
1	1			
2	2			
4	4			
5	5			
6				
7				
8	15		=TEILERGEBNIS(9;A1:A5)	
9	12		=TEILERGEBNIS(109;A1:A5)	
10				

Abbildung 7.4: TEILERGEBNIS und ausgeblendete Zeilen

D28		fx =TEILERGEBNIS(109;E\$3:E\$26)						
	A	B	C	D	E	F	G	H
1	Name	Vorname	Geschlecht	Abteilung	Gehalt	Geburtsdatum		
3	Binsen	Jakob	m	BH	1.918,58 €	11.12.1956		
8	Hark	Grete	w	BH	2.594,39 €	01.08.1952		
10	Klein	David	m	BH	1.018,58 €	20.05.1960		
15	Moßner	Ursula	w	BH	2.377,26 €	16.12.1952		
22	Thomann	Albert	m	BH	1.078,69 €	12.03.1958		
27								
28	Gehaltssumme Abteilung BH			8.987,50 €		=TEILERGEBNIS(109;E\$3:E\$26)		
29								

Abbildung 7.5: Gehaltssumme der Abteilung BH mit TEILERGEBNIS



Es ist wichtig, dass alle Auswertungen, wie Berechnungen, Ergebnisse oder kopierte Ergebnisse von Spezialfiltern entweder oberhalb, unterhalb der Datengrundlage oder auf einem eigenen Tabellenblatt platziert wird.

Es kann sonst passieren, dass durch das Filter diese Ergebnisse mit ausgeblendet werden und daher nicht mehr sichtbar sind!

7.3 Spezialfilter

Mit dem Autofilter stoßen wir recht bald an die Grenzen der Übersichtlichkeit und Flexibilität. Das Spezialfilter ist ein mächtigeres Autofilter, bei dem die Filterkriterien explizit angeführt werden müssen.

Kriterienbereich

↓

Name	Vorname	Geschlecht	Abteilung	Gehalt	Gehalt	Geburtsdatum
			BH	>2000	<3000	<1.1.1964
			VK	>=1500	<2500	<1.1.1964

Name	Vorname	Geschlecht	Abteilung	Gehalt	Geburtsdatum
Atzenegger	Ingo	m	EK	1.415,28 €	06.09.1953
Binsen	Jakob	m	BH	1.918,58 €	11.12.1956
Braatz	Uwe	m	VK	1.398,90 €	21.11.1961
Braun	Quirina	w	DP	2.744,97 €	13.01.1954
Gründel	Ursula	w	VK	1.465,02 €	09.05.1960
Hansen	Petra	w	VK	2.764,94 €	16.08.1966
Hark	Grete	w	BH	2.594,39 €	01.08.1952
Heinicke	Clara	w	DP	1.901,44 €	27.08.1951
Klein	David	m	BH	1.018,58 €	20.05.1960
Klein	Nora	w	VK	2.896,53 €	17.01.1969
Konrad	Grete	w	EK	1.231,36 €	24.12.1961
Kurtz	Xaver	m	DV	1.874,47 €	09.03.1968
Meier	Quirina	w	EK	2.131,15 €	30.08.1957
Mößner	Ursula	w	BH	2.377,26 €	16.12.1952
Muscheid	David	m	DP	2.557,47 €	15.12.1950
Saftig	Emma	w	DV	2.014,58 €	28.05.1953
Saftig	Friederike	w	EK	2.789,16 €	24.11.1968
Schreiber	Raffaela	w	DV	2.893,72 €	14.05.1958
Schultz	Xaver	m	DP	1.164,36 €	04.05.1956
Seeler	Xenia	w	DV	2.650,63 €	18.08.1956
Thomann	Albert	m	BH	1.078,69 €	12.03.1958
Traunert	Petra	w	EK	2.833,94 €	07.11.1957
Trautner	Stefan	m	VK	1.372,57 €	08.12.1959
Volker	Ingrid	w	DP	1.721,36 €	17.02.1961
Wersing	Uwe	m	EK	1.107,69 €	02.12.1958

Listenbereich (Datengrundlage)

←

Abbildung 7.6: Spezialfilter mit Datenbereich und Kriterienbereich

7.4 Spezialfilter Kriterienbereich

Der Kriterienbereich des Spezialfilters wird benötigt, um Regeln zu erstellen, welche Datensätze gefiltert werden sollen

Name	Vorname	Geschlecht	Abteilung	Gehalt	Gehalt	Geburtsdatum
			BH	>2000	<3000	<1.1.1964
			VK	>=1500	<2500	<1.1.1964

Abbildung 7.7: Spezialfilter Kriterienbereich

Wie funktioniert nun das Filtern mit einem Spezialfilter? Vom Listenbereich kopiert man jene Überschriften, nach deren Wertinhalt gefiltert werden soll. Diese Überschriften kopiert man in einen Bereich oberhalb oder unterhalb des Listenbereiches. Soll in einer Spalte nach einem Bereich gefiltert werden, so muss die Überschrift zwei Mal abgebildet werden. Einmal für den Vergleich für die untere Grenze und ein Mal für die obere Grenze.

In oben gezeigtem Beispiel werden alle MitarbeiterInnen gefiltert, welche vor dem 1.1.1964 geboren wurden, in der Abteilung BH arbeiten und ein Gehalt von über 2.000 und unter 3.000 Euro erhalten. Des weiteren werden auch Datensätze von Mitarbeitenden angezeigt, welche vor dem 1.1.1964 geboren wurden, in der Abteilung VK arbeiten und ein Gehalt von mindestens 1.500 und unter 2.500 Euro verdienen.



Merke:

- Alle Spalten in einer Kriterienzeile sind *UND* verknüpft
- Alle Zeilen des Kriterienbereiches sind *ODER* verknüpft



Es ist immer besser, wenn man die Spaltenüberschriften kopiert und nicht tippt, da sich Tippfehler einschleichen können. Tippfehler bewirken nämlich nicht, dass das Filtern nicht funktioniert, sondern man erhält ein berechnetes Feld, welches in Kapitel 7.7 noch näher erklärt wird.

7.5 Spezialfilter Vergleichsoperationen

Vergleichsoperatoren für numerische- oder Datumswerte	
ist gleich 20	= 20
größer 20	> 20
größer gleich 20	>= 20
kleiner 20	< 20
kleiner gleich 20	<= 20
ungleich 20	<> 20

Tabelle 7.2: Numerische und Datumsvergleichsoperatoren

Vergleichsoperatoren für alphanumerische Werte	
alle die mit „Klein“ beginnen	Klein
alle die im Text „Klein“ beinhalten	*Klein*
alle Werte die mit A, B oder C beginnen (D ist ausgeschlossen)	< D
alle Werte die mit X, Y oder Z beginnen (Xx ist eingeschlossen)	> X
beliebiges Zeichen an einer bestimmten Stelle	?
kein, ein oder mehrere beliebige Zeichen	*
„*“ als normales Zeichen	~*
„?“ als normales Zeichen	~?
„~“ als normales Zeichen	~~
Suchen nach exaktem Wert unter Berücksichtigung der Groß- und Kleinschreibung	=IDENTISCH(A5; "Klein")
Suche nach leerer Zelle	=""

Tabelle 7.3: Alphanumerische Vergleichsoperatoren

7.6 Spezialfilter aktivieren

Bevor man das Spezialfilter aktiviert, sollte man sich überlegen, wo die gefilterten Daten angezeigt werden sollen. Es gibt hierbei drei Möglichkeiten.

- An der gleichen Stelle wie die Datengrundlage anzeigen
- Am selben Tabellenblatt wie die Datengrundlage anzeigen, aber an anderer Stelle
- Auf einem anderen Tabellenblatt anzeigen

Die Vorgehensweise ist dabei identisch, es wird mit **Daten** **Sortieren und Filtern** **Erweitert** aufgerufen




Abbildung 7.8: Spezialfilter Menü

Danach wird in der Dialogbox entweder *An gleicher Stelle filtern* gewählt, wenn die Daten im Listenbereich angezeigt werden sollen. Will man dagegen die Daten in einem anderen Bereich eines Arbeitsblattes oder in einem anderen Arbeitsblatt anzeigen, dann muss man *An eine andere Stelle kopieren* auswählen




Abbildung 7.9: Spezialfilter aktivieren



Will man die gefilterten Daten am gleich Datenblatt haben, so empfiehlt es sich, dass man vor dem Aufrufen des Spezialfilters den Cursor innerhalb des Datenbereiches setzt, da dieser dann automatisch von Excel erkannt und ausgewählt wird.



Eine besondere Falle liefert Excel, wenn das Ergebnis auf ein anderes Tabellenblatt kopiert werden soll. Man muss *immer* zwingend von dem Tabellenblatt ausgehen, auf dem die gefilterten Daten angezeigt werden. Macht man dies nicht, so liefert Excel die Fehlermeldung




Abbildung 7.10: Spezialfilter Fehlermeldung

7.7 Spezialfilter mit berechneten Feldern

Eine Besonderheit im Kriterienbereich sind berechnete Felder. Im Gegensatz zu normalen Kriterien wird hier nicht direkt eine Spalte der Datengrundlage mit einem Wert verglichen, sondern der zu vergleichende Wert wird über eine Formel bestimmt. Berechnete Felder kommen zum Einsatz, wenn an der Datengrundlage nichts geändert werden darf oder kann.

Nehmen wir als Beispiel, dass wir alle MitarbeiterInnen der Abteilung BH herausfinden wollen, welche ein Jahreseinkommen von über 30.000€ haben. Das Jahreseinkommen ist das 14-fache des Monatsgehaltes plus der Zulage.

Bei berechneten Feldern sind folgende Dinge zu beachten

- Die Überschrift im Kriterienbereich darf *niemals* mit einer der Überschriften der Datengrundlage übereinstimmen. Der Grund liegt darin, dass es sich hier um eine „künstliche“ Spalte handelt.
- Berechnete Felder beginnen immer mit =, da es sich um eine Formel handelt
- Die Formel in berechneten Feldern muss immer WAHR oder FALSCH ergeben
- Wenn bei der Berechnung Werte der Datengrundlage herangezogen werden, muss *immer* der Wert aus der *ersten* Zeile der Datengrundlage genommen werden.

Wie wird das nun in der zu lösenden Aufgabe umgesetzt?

E2		fx		=14*E5+G5>30000			
	A	B	C	D	E	F	G
1	Name	Vorname	Geschlecht	Abteilung	Einkommen		
2				BH	FALSCH		
3							
4	Name	Vorname	Geschlecht	Abteilung	Gehalt	Geburtsdatum	Zulage
5	Atzenegger	Ingo	m	EK	1.415,28 €	06.09.1953	1.500,00 €
6	Binsen	Jakob	m	BH	1.918,58 €	11.12.1956	1.500,00 €

Abbildung 7.11: Spezialfilter mit berechnetem Feld

Die Formel für die Berechnung lautet $14 * Gehalt + Zulage$. Mit dieser Formel soll jede Zeile des Listenbereichs verglichen werden. Ist das Ergebnis größer als 30.000, dann soll diese Zeile im Ergebnis aufscheinen. Die erste Zeile des Listenbereiches ist die Zeile 5, also muss diese für den Vergleich herangezogen werden. Das Kriterium lautet daher

$$= 14 * E5 + G5 > 30000 \quad (1)$$

wie auch in der Abbildung ersichtlich ist.

8 Datenbankfunktionen

Datenbankfunktionen ermöglichen eine aggregierte Auswertung großer Listen mit komplexen Filterkriterien, die identisch aufgebaut sind, wie jene des Spezialfilter Kriterienbereiches. Deswegen sie sehr oft in Zusammenhang mit dem Spezialfilter verwendet, da sie neben dem reinen Filtern von Datensätzen auch wichtige Informationen wie Maximum Minimum oder Summe eines gefilterten Datenbereiches liefern. Im Prinzip kann da auch die bereits bekannte TEILERGEBNIS Funktion, welche aber den Nachteil hat, dass bei geänderten Bedingungen *keine* Neuberechnung stattfindet.

Die Datenbankfunktionen sind

- DBANZAHL
- DBANZAHL2
- DBMAX
- DBMIN
- DBPRODUKT
- DBMITTELWERT
- DBSTDABWN

Die Anzahl und der Aufbau der Parameter der Datenbankfunktionen sind bei alles oben genannten gleich, daher folgt hier die Beschreibung anhand von DBSUMME.

$$\text{DBSUMME}(\langle \text{Datenbank} \rangle; \langle \text{Datenbankfeld} \rangle; \langle \text{Suchkriterien} \rangle)$$

- ⟨Datenbank⟩ Ist die Datengrundlage *inklusive* der Überschriften
- ⟨Datenbankfeld⟩ Ist die Spalte an der die Datenbankfunktion durchgeführt werden soll. Wichtig ist hier, dass die Überschrift der Spalte ausgewählt wird.
- ⟨Suchkriterien⟩ Diese bestimmen, genauso wie beim Kriterienbereich des Spezialfilters, welche Datensätze für die Aggregatfunktion herangezogen werden sollen. Wie auch beim Spezialfilter werden im Kriterienbereich Spalten UND verknüpft und Zeilen ODER verknüpft.

Als Beispiel will ein Controller folgende Dinge in einem Unternehmen wissen

- wie viele Mitarbeiter arbeiten in der Abteilung VK,
- wie viele Mitarbeiter sind davon über 50 Jahre alt,
- die Gehaltssumme der Abteilung VK

	A	B	C	D	E	F	G	H
1	Name	Vorname	Abteilung	Alter>50				
2			VK	WAHR		=DATEDIF(F11;HEUTE();"y")>=50		
3								
4	Anzahl Mitarbeiter VK			5		=DBANZAHL2(\$A\$10:\$F\$35;A10;\$C\$1:\$C\$2)		
5	Mitarbeiter VK >= 50 Jahre			3		=DBANZAHL2(\$A\$10:\$F\$35;A10;\$A\$1:\$D\$2)		
6	Gehaltssumme VK			9.897,96 €		=DBSUMME(\$A\$10:\$F\$35;E10;\$C\$1:\$C\$2)		
7								
8								
9								
10	Name	Vorname	Geschlecht	Abteilung	Gehalt	Geburtsdatum		
13	Braatz	Uwe	m	VK	1.398,90 €	21.11.1961		
15	Gründel	Ursula	w	VK	1.465,02 €	09.05.1960		
16	Hansen	Petra	w	VK	2.764,94 €	16.08.1966		
20	Klein	Nora	w	VK	2.896,53 €	17.01.1969		
33	Trautner	Stefan	m	VK	1.372,57 €	08.12.1959		
36								

Abbildung 8.1: Datenbankfunktionen

Will der Controller nun statt der Abteilung VK jetzt die Abteilung EK wissen, braucht er nur in der Zelle C2 den Wert "VK" auf "EK" ändern und erhält sofort das Ergebnis.

9 Erweiterte WENN Funktionen

9.1 SUMMEWENN Funktion

Mit dieser Funktion kann man in Zellbereichen numerische Werte addieren, die bestimmten Bedingungen in gleichen oder anderen Zellbereichen unterworfen sind.

WENN(⟨Bereich⟩; ⟨Suchkriterien⟩; ⟨SUMME_BEREICH⟩)

Das folgende Beispiel zeigt die Umsätze in Abhängigkeit der Kategorie

B11		fx =SUMMEWENN(\$C\$2:\$C\$8;B10;\$B\$2:\$B\$8)					
	A	B	C	D	E	F	G
1	Art	Umsatz	Kategorie				
2	Birne	3.450,00 €	1				
3	Banane	5.623,00 €	1				
4	Apfel	8.211,00 €	1				
5	Käse	4.356,00 €	2				
6	Joghurt	8.200,00 €	2				
7	Salami	2.345,00 €	3				
8	Schinken	6.712,00 €	3				
9							
10	Kategorie	2					
11	Umsatz	12.556,00 €					
12							

Abbildung 9.1: SUMMEWENN Funktion

9.2 SUMMEWENNS Funktion

Bei dieser Funktion kann man Werte addieren, welche anhand von bis zu 127 Kriterien gefiltert wurden

WENN(⟨Summe_Bereich⟩; ⟨Kriterien_Bereich1⟩; ⟨Kriterien1⟩; . . .)

⟨Summe_Bereich⟩ Jener Zellbereich, der summiert werden soll.

⟨Kriterien_Bereich1⟩ Jener Zellbereich, auf den das Kriterium (n) angewendet werden soll.

⟨Kriterien1⟩ Jene Suchbedingung, die auf den Kriterien Bereich (n) angewendet werden soll.

Abbildung 9.2 zeigt die Umsätze in Abhängigkeit der Kategorie und eines Zeitraumes

F5		fx =SUMMEWENNS(B:B;A:A;">=" & E3;A:A;"<=" & F3;C:C;G3)							
	A	B	C	D	E	F	G	H	I
1	Datum	Umsatz	Kategorie			Auswertung			
2	01.07.2015	7.625,00 €	1		von	bis	Kategorie		
3	01.07.2015	1.222,00 €	2		01.07.2015	03.07.2015	1		
4	02.07.2015	2.762,00 €	1						
5	02.07.2015	6.590,00 €	2		Umsatz	12.187,00 €			
6	03.07.2015	1.800,00 €	1						
7	03.07.2015	7.599,00 €	2						
8	04.07.2015	4.872,00 €	1						

Abbildung 9.2: SUMMEWENNS Funktion

9.3 ZÄHLENWENN Funktion

Mit dieser Funktion kann man die Anzahl der zutreffenden Zellen zählen, die bestimmten Bedingungen entsprechen.

$\text{WENN}(\langle \text{Bereich} \rangle; \langle \text{Suchkriterien} \rangle; \langle \text{SUMME_BEREICH} \rangle)$

Das folgende Beispiel zeigt die Umsätze in Abhängigkeit der Kategorie


B11			f_x	=ZÄHLENWENN(\$C\$2:\$C\$8;B10)		
	A	B	C	D	E	F
1	Art	Umsatz	Kategorie			
2	Birne	3.450,00 €	1			
3	Banane	5.623,00 €	1			
4	Apfel	8.211,00 €	1			
5	Käse	4.356,00 €	2			
6	Joghurt	8.200,00 €	2			
7	Salami	2.345,00 €	3			
8	Schinken	6.712,00 €	3			
9						
10	Kategorie	1				
11	Anzahl	3		=ZÄHLENWENN(\$C\$2:\$C\$8;B10)		
12						

Abbildung 9.3: ZÄHLENWENN Funktion

9.4 ZÄHLENWENNS Funktion

Bei dieser Funktion kann man Werte zählen, welche anhand von bis zu 127 Kriterien gefiltert wurden

$\text{WENN}(\langle \text{Summe_Bereich} \rangle; \langle \text{Kriterien_Bereich1} \rangle; \langle \text{Kriterien1} \rangle; . . .)$

$\langle \text{Summe_Bereich} \rangle$ Jener Zellbereich, der summiert werden soll.

$\langle \text{Kriterien_Bereich1} \rangle$ Jener Zellbereich, auf den das Kriterium (n) angewendet werden soll.

$\langle \text{Kriterien1} \rangle$ Jene Suchbedingung, die auf den Kriterien Bereich (n) angewendet werden soll.

Abbildung 9.4 zeigt die Anzahl von Personen, welche in einem bestimmter Altersbereich sind, Raucher sind und Übergewicht haben

G5 fx =ZÄHLENWENNS(B:B;F3;B:B;G3;C:C;H3;D:D;I3)									
	A	B	C	D	E	F	G	H	I
1	Person	Alter	Raucher	Übergewicht	Auswertung				
2	Atzenegger	17	✓	✓		Alter	Alter	Raucher	Übergewicht
3	Binsen	21	✓	✗		>15	<25	1	1
4	Braatz	38	✓	✓					
5	Braun	24	✗	✗		Anzahl	2		
6	Gründel	36	✓	✗					
7	Hansen	18	✓	✓		=ZÄHLENWENNS(B:B;F3;B:B;G3;C:C;H3;D:D;I3)			
8	Hark	64	✗	✗					
9	Heinicke	57	✓	✓					
10	Klein	44	✗	✓					
11	Klein	19	✓	✗					
12	Konrad	68	✗	✓					

Abbildung 9.4: ZAEHLENWENNS Funktion